

# Best Title in the Universe

42

Victor Cacciari Miraldo    Wouter Swierstra

University of Utrecht

{v.cacciarimaldo,w.s.swierstra} at uu.nl

## Abstract

stuff

**Categories and Subject Descriptors** D.1.1 [look]: for—this

**General Terms** Haskell

**Keywords** Haskell

## 1. Introduction

The majority of version control systems handle patches in a non-structured way. They see a file as a list of lines that can be inserted, deleted or modified, with no regard to the semantics of that specific file. The immediate consequence of such design decision is that we, humans, have to solve a large number of conflicts that arise from, in fact, non conflicting edits. Implementing a tool that knows the semantics of any file we happen to need, however, is no simple task, specially given the plethora of file formats we see nowadays.

This can be seen from a simple example. Lets imagine Alice and Bob are iterating over a cake’s recipe. They decide to use a version control system and an online repository to keep track of their modifications.




Figure 1. Sample CSV files

Lets say that both Bob and Alice are happy with their independent changes and want to make a final recipe. The standard way to track differences between files is the `diff3` [FIXBIB] unis tool. Running `diff3 Alice.csv 0.csv Bob.csv` would result in the output presented in figure 2. Every tag `====` marks a difference. Three

locations follows, formatted as `file:line type`. The change type can be a *Change*, *Append* or *Delete*. The first one, says that file 1 (Alice.csv) has a change in line 2 (1:2c) which is `flour, 2 , cp`; and files 2 and 3 have different changes in the same line. The tag `====3` indicates that there is a difference in file 3 only. Files 1 and 2 should append what changed in file 3 (line 4).

```
====
1:2c
    flour, 2 , cp
2:2c
    flour, 1 , cp
3:2c
    cakeflour, 1 , cp
====3
1:3a
2:3a
3:4c
    sugar, 1 , tsp
```

Figure 2. Output from `diff3`

If we try to merge the changes, `diff3` will flag a conflict and therefore require human interaction to solve it, as we can see by the presence of the `====` indicator in its output. However, Alice’s and Bob’s edits, in figure 1 do *not* conflict, if we take into account the semantics of CSV files. Although there is an overlapping edit at line 1, the fundamental editing unit is the cell, not the line.

We propose a structural diff that is not only generic but also able to track changes in a way that the user has the freedom to decide which is the fundamental editing unit. Our work was inspired by [5] and [10]. We did extensive changes in order to handle structural merging of patches. We also propose extensions to this algorithm capable of detecting purely structural operations such as refactorings and cloning.

The paper begins by exploring the problem, generically, in the Agda [FIXBIB] language. Once we have a provably correct algorithm, the details of a Haskell implementation of generic diff’ing are sketched. To open ground for future work, we present a few extensions to our initial algorithm that could be able to detect semantical operations such as *cloning* and *swapping*.

## Contributions

- We provide a type-indexed definition of a generic diff, over the universe of context-free types.
- We provide the base, proven to be correct, algorithms for computing and applying a diff generically.

Permission to make digital or hard copies of part or all of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than ACM must be honored. Abstracting with credit is permitted. To copy otherwise, to republish, to post on servers, or to redistribute to lists, contact the Owner/Author. Request permissions from permissions@acm.org or Publications Dept., ACM, Inc., fax +1 (212) 869-0481. Copyright held by Owner/Author. Publication Rights Licensed to ACM.

- A notion of residual is used to propagate changes of different diffs, hence providing a bare mechanism for merging and conflict resolution.
- We provide a Haskell prototype with advanced, user defined, automatic conflict solving strategies.

## Background

- *Should we have this section? It could be nice to at least mention the edit distance problem and that in the untyped scenario, the best running time is of  $O(n^3)$ . Types should allow us to bring this time lower.*


## 2. Structural Diffing

Alice and Bob were both editing a CSV file which represents data that is isomorphic to  $[[Atom\ String]]$ , where  $Atom\ a$  is a simple tag that indicates that  $as$  should be treated abstractly, that is, either they are equal or different, we will not open these values to check for structural changes.

As we are tracking differences, there are a few operations that are inherent to our domain, such as: inserting; deleting; copying and updating. When we say *structural diffing*, however, we add another option to this list. Now we will also be able to go down the structure of some object and inspect its parts. To illustrate this, let us take Alice's change as in figure 1, her changes to the file could be described, structurally, as:

- I) Copy the first line;
- II) Enter the second line;
  - i) Copy the first field;
  - ii) Enter the second field;
    - Update atom "1" for atom "2".
  - iii) Copy the third field;
- III) Copy the third line.
- IV) Finish.

In figure 3 we show the patch that corresponds to that.



**Figure 3.** Alice's Patch

Consider now Bob's structural changes to the CSV file<sup>1</sup>. If you overlap both, you should notice that there is *Upd* operation on top of another. This was in fact expected given that Alice and Bob performed changes in disjoint parts of the CSV file.

- *Diffing and tree-edit distance are very closely related problems.*
- *This should go on background, though.*

### To Research!

<sup>1</sup>Exercise to the reader! Clue: the last two operations are *Ins* ["sugar", "1", "tsp"] *End*

- *The LCS problem is closely related to diffing. We want to preserve the LCS of two structures! How does our diffing relate? Does this imply maximum sharing?*

- *ANS: No! We don't strive for maximum sharing. We strive for flexibility and customization. See refactoring*

- *connect this section and the next*

### 2.1 Context Free Datatypes

Although our running example, of CSV files, has type  $[[Atom\ String]]$ , lists of  $a$  themselves are in fact the least fixed point of the functor  $X \mapsto 1 + a \times X$ . Which is a *context-free type*, in the sense of [1]. For it is constructed following the grammar CF of context free types with a deBruijn representation for variables.

$$CF ::= 1 \mid 0 \mid CF \times CF \mid CF + CF \mid \mu CF \mid \mathbb{N}$$

In Agda, the CF universe is defined by:

```
data U : N → Set where
  u0 : {n : N} → U n
  u1 : {n : N} → U n
  _⊕_ : {n : N} → U n → U n → U n
  _⊗_ : {n : N} → U n → U n → U n
  β  : {n : N} → U (suc n) → U n → U n
  μ  : {n : N} → U (suc n) → U n
  vl : {n : N} → U (suc n)
  wk : {n : N} → U n → U (suc n)
```

Here,  $\beta$  stands for type application;  $vl$  is the topmost variable in scope and  $wk$  ignores the topmost variable in scope. We could have used a *Fin* to identify variables, and have one instead of two constructors for variables, but that would trigger more complicated definitions later on.

We stress that one of the main objectives of this project is to release a solid diffing and merging tool, that can provide formal guarantees, written in Haskell. The universe of user-defined Haskell types is smaller than context free types; in fact, we have fixed-points of sums-of-products. Therefore, we should be able to apply the knowledge acquired in Agda directly in Haskell. In fact, we did so! With a few adaptations here and there, to make the type-checker happy, the Haskell code is almost a direct translation, and will be discussed in section 5.

Stating the language of our types is not enough. We need to specify its elements too, after all, they are the domain which we seek to define our algorithms for! Defining elements of fixed-point types make things a bit more complicated, check [1] for a more in-depth explanation of these details. Long story short, we have to use a decreasing *Telescope* to satisfy the termination checker. In Agda, the elements of  $U$  are defined by:

```

data EIU : {n : ℕ} → U n → Tel n → Set where
  void : {n : ℕ} {t : Tel n}
    → EIU u1 t
  inl   : {n : ℕ} {t : Tel n} {a b : U n}
    (x : EIU a t) → EIU (a ⊕ b) t
  inr   : {n : ℕ} {t : Tel n} {a b : U n}
    (x : EIU b t) → EIU (a ⊕ b) t
  _,_   : {n : ℕ} {t : Tel n} {a b : U n}
    → EIU a t → EIU b t → EIU (a ⊗ b) t
  top   : {n : ℕ} {t : Tel n} {a : U n}
    → EIU a t → EIU v1 (tcons a t)
  pop   : {n : ℕ} {t : Tel n} {a b : U n}
    → EIU b t → EIU (wk b) (tcons a t)
  mu     : {n : ℕ} {t : Tel n} {a : U (suc n)}
    → EIU a (tcons (μ a) t) → EIU (μ a) t
  red    : {n : ℕ} {t : Tel n} {F : U (suc n)} {x : U n}
    → EIU F (tcons x t)
    → EIU (β F x) t

```

The `Tel` index is the telescope in which to look for the instantiation of type-variables. A value  $(v : \text{EIU } \{n\} \text{ ty } t)$  reads roughly as: a value of type  $\text{ty}$  with  $n$  variables, applied to  $n$  types  $t$  with at most  $n - 1$  variables. We need this decrease of type variables to convince the termination checker that our code is ok. It's Agda definition is:

```

data Tel : ℕ → Set where
  tnil : Tel 0
  tcons : {n : ℕ} → U n → Tel n → Tel (suc n)

```

Let us see a simple example of how types and elements are defined in this framework. Consider that we want to encode the list  $(u : []) :: [U]$ , for  $U$  being the unit type with the single constructor  $u$ . We start by defining the type of lists, this is an element of  $U$  (`suc n`), which later lets us define an element of that type.

```

list : {n : ℕ} → U (suc n)
list = μ (u1 ⊕ wk v1 ⊗ v1)

myList : {n : ℕ} {t : Tel n} → EIU list (tcons u1 t)
myList = mu (inr (pop (top void) , top (mu (inl void))))

```

So far so good. We seem to have the syntax figured out. But which operations can we perform to these elements? As we shall see, this choice of universe turns out to be very expressive, providing a plethora of interesting operations. The first very useful concept is the decidability of generic equality[1].

```

_≐_ : {n : ℕ} {t : Tel n} {u : U n} (x y : EIU u t) → Dec (x ≐ y)

```

But only comparing things will not get us very far. We need to be able to inspect our elements generically. Things like getting the list of immediate children, or computing their arity, that is, how many children do they have, are very useful.

```

children : {n : ℕ} {t : Tel n} {a : U (suc n)} {b : U n}
  → EIU a (tcons b t) → List (EIU b t)

arity : {n : ℕ} {t : Tel n} {a : U (suc n)} {b : U n}
  → EIU a (tcons b t) → ℕ

```

The advantage of doing so in Agda, is that we can prove that our definitions are correct.

```

children-arity-lemma
  : {n : ℕ} {t : Tel n} {a : U (suc n)} {b : U n}
  → (x : EIU a (tcons b t))
  → length (children x) ≐ arity x

```

We can even go a step further and say that every element is defined by a constructor and a vector of children, with the correct arity. This lets us treat generic elements as elements of a (typed) rose-tree, whenever this is convenient.

```

unplug : {n : ℕ} {t : Tel n} {a : U (suc n)} {b : U n}
  → (el : EIU a (tcons b t))
  → Σ (EIU a (tcons u1 t)) (λ x → Vec (EIU b t) (arity x))

plug : {n : ℕ} {t : Tel n} {a : U (suc n)} {b : U n}
  → (el : EIU a (tcons u1 t))
  → Vec (EIU b t) (arity el)
  → EIU a (tcons b t)

plug-correct : {n : ℕ} {t : Tel n} {a : U (suc n)} {b : U n}
  → (el : EIU a (tcons b t))
  → el ≐ plug (p1 (unplug el)) (p2 (unplug el))

```

#### • Vassena's and Loh's universe is the typed rose-tree! Correlate!!

This repertoire of operations, and the hability to inspect an element structurally, according to its type, gives us the toolset we need in order to start describing differences between elements. That is, we can now start discussing what does it mean to *diff* two elements or *patch* an element according to some description of changes.

## 2.2 Patches over a Context Free Type

A patch over  $T$  is an object that describe possible changes that can be made to objects of type  $T$ . The high-level idea is that diffing two objects  $t_1, t_2 : T$  will produce a patch over  $T$ , whereas applying a patch over  $A$  to an object will produce a *Maybe T*. It is interesting to note that application can not be made total. Let's consider  $T = X + Y$ , and now consider a patch  $(Left\ x) \xrightarrow{p} (Left\ x')$ . What should be the result of applying  $p$  to a  $(Right\ y)$ ? It is undefined!

The type of *diff*'s is defined by `D`. It is indexed by a type and a telescope, which is the same as saying that we only define *diff*'s for closed types<sup>2</sup>. However, it also has a parameter  $A$ , this will be addressed later.

```

data D {a} (A : {n : ℕ} → Tel n → U n → Set a)
  : {n : ℕ} → Tel n → U n → Set a where

```

As we mentioned earlier, we are interested in analyzing the set of possible changes that can be made to objects of a type  $T$ . These changes depend on the structure of  $T$ , for the definition follows by induction on it.

For  $T$  being the Unit type, we can not modify it.

```

D-void : {n : ℕ} {t : Tel n} → D A t u1

```

For  $T$  being a product, we need to provide *diffs* for both its components.

```

D-pair : {n : ℕ} {t : Tel n} {a b : U n}
  → D A t a → D A t b → D A t (a ⊗ b)

```

For  $T$  being a coproduct, things become slightly more interesting. There are four possible ways of modifying a coproduct, which are defined by:

<sup>2</sup>Types that do not have any free type-variables

$\text{D-inl} : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{a b : \mathbb{U} n\}$   
 $\rightarrow \text{D } A \text{ } t a \rightarrow \text{D } A \text{ } t (a \oplus b)$   
 $\text{D-inr} : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{a b : \mathbb{U} n\}$   
 $\rightarrow \text{D } A \text{ } t b \rightarrow \text{D } A \text{ } t (a \oplus b)$   
 $\text{D-setl} : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{a b : \mathbb{U} n\}$   
 $\rightarrow \text{E!U } a \text{ } t \rightarrow \text{E!U } b \text{ } t \rightarrow \text{D } A \text{ } t (a \oplus b)$   
 $\text{D-setr} : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{a b : \mathbb{U} n\}$   
 $\rightarrow \text{E!U } b \text{ } t \rightarrow \text{E!U } a \text{ } t \rightarrow \text{D } A \text{ } t (a \oplus b)$

We also need some housekeeping definitions to make sure we handle all types defined by  $\mathbb{U}$ .

$\text{D-}\beta : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{F : \mathbb{U} (\text{succ } n)\} \{x : \mathbb{U} n\}$   
 $\rightarrow \text{D } A \text{ } (t \text{cons } x \text{ } t) F \rightarrow \text{D } A \text{ } t (\beta F x)$   
 $\text{D-top} : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{a : \mathbb{U} n\}$   
 $\rightarrow \text{D } A \text{ } t a \rightarrow \text{D } A \text{ } (t \text{cons } a \text{ } t) \text{v!}$   
 $\text{D-pop} : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{a b : \mathbb{U} n\}$   
 $\rightarrow \text{D } A \text{ } t b \rightarrow \text{D } A \text{ } (t \text{cons } a \text{ } t) (\text{wk } b)$

Fixed points are handled by a list of *edit operations*. We will discuss them in detail later on.

$\text{D-mu} : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{a : \mathbb{U} (\text{succ } n)\}$   
 $\rightarrow \text{List } (\text{D}\mu A \text{ } t a) \rightarrow \text{D } A \text{ } t (\mu a)$

The aforementioned parameter  $A$  goes is used in a single constructor, allowing us to have a free-monad structure over  $\text{D}$ 's. This shows to be very usefull for adding extra information, as we shall discuss, on section 3.1, for adding conflicts.

$\text{D-A} : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{ty : \mathbb{U} n\} \rightarrow A \text{ } t \text{ } ty \rightarrow \text{D } A \text{ } t \text{ } ty$

Finally, we define  $\text{Patch } t \text{ } ty$  as  $\text{D } (\lambda \_ \_ \rightarrow \perp) t \text{ } ty$ . Meaning that a  $\text{Patch}$  is a  $\text{D}$  with *no* extra information.

### 2.3 Producing Patches

Given a generic definition of possible changes, the primary goal is to produce an instance of this possible changes, for two specific elements of a type  $T$ . We shall call this process *diffing*. It is important to note that our  $\text{gdiff}$  function expects two elements of the same type! This contrasts with the work done by Vassena[10] and Lempink[5], where their  $\text{diff}$  takes objects of two different types.

For types which are not fixed points, the  $\text{gdiff}$  functions looks like:

$\text{gdiff} : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{ty : \mathbb{U} n\}$   
 $\rightarrow \text{E!U } ty \text{ } t \rightarrow \text{E!U } ty \text{ } t \rightarrow \text{Patch } t \text{ } ty$   
 $\text{gdiff} \{ty = \text{v!}\} (\text{top } a) (\text{top } b) = \text{D-top } (\text{gdiff } a \text{ } b)$   
 $\text{gdiff} \{ty = \text{wk } u\} (\text{pop } a) (\text{pop } b) = \text{D-pop } (\text{gdiff } a \text{ } b)$   
 $\text{gdiff} \{ty = \beta F x\} (\text{red } a) (\text{red } b) = \text{D-}\beta (\text{gdiff } a \text{ } b)$   
 $\text{gdiff} \{ty = \text{u1}\} \text{void void} = \text{D-void}$   
 $\text{gdiff} \{ty = ty \otimes tv\} (ay, av) (by, bv)$   
 $= \text{D-pair } (\text{gdiff } ay \text{ } by) (\text{gdiff } av \text{ } bv)$   
 $\text{gdiff} \{ty = ty \oplus tv\} (\text{inl } ay) (\text{inl } by) = \text{D-inl } (\text{gdiff } ay \text{ } by)$   
 $\text{gdiff} \{ty = ty \oplus tv\} (\text{inr } av) (\text{inr } bv) = \text{D-inr } (\text{gdiff } av \text{ } bv)$   
 $\text{gdiff} \{ty = ty \oplus tv\} (\text{inl } ay) (\text{inr } bv) = \text{D-setl } ay \text{ } bv$   
 $\text{gdiff} \{ty = ty \oplus tv\} (\text{inr } av) (\text{inl } by) = \text{D-setr } av \text{ } by$   
 $\text{gdiff} \{ty = \mu ty\} a \text{ } b = \text{D-mu } (\text{gdiffL } (a :: []) (b :: []))$

Where the  $\text{gdiffL}$  takes care of handling fixed point values. The important remark here is that it operates over lists of elements, instead of single elements. This is due to the fact that the children of a fixed point element is a (possibly empty) list of fixed point elements.

**Fixed Points** have a fundamental difference over regular algebraic datatypes. They can grow or shrink arbitralily. We have to account for that when tracking differences between their elements.

As we mentioned earlier, the  $\text{diff}$  of a fixed point is defined by a list of *edit operations*.

$\text{data D}\mu \{a\} (A : \{n : \mathbb{N}\} \rightarrow \text{Tel } n \rightarrow \mathbb{U} n \rightarrow \text{Set } a)$   
 $: \{n : \mathbb{N}\} \rightarrow \text{Tel } n \rightarrow \mathbb{U} (\text{succ } n) \rightarrow \text{Set } a \text{ where}$

Again, we have a constructor for adding *extra* information, which is ignored in the case of  $\text{Patches}$ .

$\text{D}\mu\text{-A} : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{a : \mathbb{U} (\text{succ } n)\}$   
 $\rightarrow A \text{ } t (\mu a) \rightarrow \text{D}\mu A \text{ } t a$

But the interesting bits are the *edit operations* we allow, where  $\text{Val } a \text{ } t = \text{E!U } a \text{ } (t \text{cons u1 } t)$ :

$\text{D}\mu\text{-ins} : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{a : \mathbb{U} (\text{succ } n)\}$   
 $\rightarrow \text{Val } a \text{ } t \rightarrow \text{D}\mu A \text{ } t a$   
 $\text{D}\mu\text{-del} : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{a : \mathbb{U} (\text{succ } n)\}$   
 $\rightarrow \text{Val } a \text{ } t \rightarrow \text{D}\mu A \text{ } t a$   
 $\text{D}\mu\text{-dwn} : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{a : \mathbb{U} (\text{succ } n)\}$   
 $\rightarrow \text{D } A \text{ } (t \text{cons u1 } t) a \rightarrow \text{D}\mu A \text{ } t a$

The reader familiar with [5] will notice that they are almost the same (adapted to our choice of universe), with two differences: our  $\text{diff}$  type is *less type-safe*, which will be discussed in section 3.2; and instead of copying, we introduce a new constructor,  $\text{D}\mu\text{-dwn}$ , which is responsible for traversing down the type-structure. Copying is modelled by  $\text{D}\mu\text{-dwn } (\text{gdiff-id } x)$ . The intuition is that for every object  $x$  there is a  $\text{diff}$  that does not change  $x$ , we will look into this on section 2.3.

Before we delve into diffing fixed point values, we need some specialization of our generic operations for fixed points. Given that  $\mu X.F \text{ } X \approx F \text{ } 1 \times [\mu X.F \text{ } X]$ , that is, any inhabitant of a fixed-point type can be seen as a non-recursive head and a list of recursive children. We then make a specialized version of the  $\text{plug}$  and  $\text{unplug}$  functions, which lets us open and close fixed point values.

$\text{Open}\mu : \{n : \mathbb{N}\} \rightarrow \text{Tel } n \rightarrow \mathbb{U} (\text{succ } n) \rightarrow \text{Set}$   
 $\text{Open}\mu \text{ } t \text{ } ty = \text{E!U } ty \text{ } (t \text{cons u1 } t) \times \text{List } (\text{E!U } (\mu ty) \text{ } t)$

$\mu\text{-open} : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{ty : \mathbb{U} (\text{succ } n)\}$   
 $\rightarrow \text{E!U } (\mu ty) \text{ } t \rightarrow \text{Open}\mu \text{ } t \text{ } ty$

$\mu\text{-close} : \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{ty : \mathbb{U} (\text{succ } n)\}$   
 $\rightarrow \text{Open}\mu \text{ } t \text{ } ty \rightarrow \text{Maybe } (\text{E!U } (\mu ty) \text{ } t \times \text{List } (\text{E!U } (\mu ty) \text{ } t))$

Although the  $\text{plug}$  and  $\text{unplug}$  uses vectors, to remain total functions, we drop that restriction and switch to lists instead, this way we can easily construct a fixed-point with the beginning of the list of children, and return the unused children, this will be very conveient when defining how patches are applied. Nevertheless, a soundness lemma guarantees the correct behaviour.

$\mu\text{-close-resp-arity}$   
 $: \{n : \mathbb{N}\} \{t : \text{Tel } n\} \{ty : \mathbb{U} (\text{succ } n)\} \{a : \text{E!U } (\mu ty) \text{ } t\}$   
 $\{hdA : \text{E!U } ty \text{ } (t \text{cons u1 } t)\} \{chA : \text{List } (\text{E!U } (\mu ty) \text{ } t)\}$   
 $\rightarrow \mu\text{-open } a \equiv (hdA, chA)$   
 $\rightarrow \mu\text{-close } (hdA, chA ++ l) \equiv \text{just } (a, l)$

We denote the first component of an *opened* fixed point by its *value*, or *head*; whereas the second component by its children. The diffing of fixed points, which was heavily inspired by [5], is then defined by:

```

gdiffL : {n : ℕ} {t : Tel n} {ty : U (suc n)}
  → List (EIU (μ ty) t) → List (EIU (μ ty) t) → Patch μ t ty
gdiffL [] [] = []
gdiffL [] (y :: ys) with μ-open y
... | hdY, chY = Dμ-ins hdY :: (gdiffL [] (chY ++ ys))
gdiffL (x :: xs) [] with μ-open x
... | hdX, chX = Dμ-del hdX :: (gdiffL (chX ++ xs) [])
gdiffL (x :: xs) (y :: ys)
= let
  hdX, chX = μ-open x
  hdY, chY = μ-open y
  d1 = Dμ-ins hdY :: (gdiffL (x :: xs) (chY ++ ys))
  d2 = Dμ-del hdX :: (gdiffL (chX ++ xs) (y :: ys))
  d3 = Dμ-dwn (gdiff hdX hdY) :: (gdiffL (chX ++ xs) (chY ++ ys))
in d1 ⊔μ d2 ⊔μ d3

```

The first three branches are simple. To transform  $[]$  into  $[]$ , we do not need to perform any action; to transform  $[]$  into  $y : ys$ , we need to insert the respective values; and to transform  $x : xs$  into  $[]$  we need to delete the respective values. The interesting case happens when we want to transform  $x : xs$  into  $y : ys$ . Here we have three possible diffs that perform the required transformation. We want to choose the diff with the least *cost* between the three of them, for we introduce an operator to do exactly that:

```

_⊔μ_ : {n : ℕ} {t : Tel n} {ty : U (suc n)}
  → Patch μ t ty → Patch μ t ty → Patch μ t ty
_⊔μ_ da db with costL da ≤? N costL db
... | yes _ = da
... | no _ = db

```

As we shall see in section 2.3.1, this notion of cost is very delicate. The idea, however, is simple. If the heads are equal we have  $d3 = Dμ-dwn (gdiff hdX hdX) \dots$ , which is how copy is implemented. We want to copy as much as possible, so we will ensure that for all  $a$ ,  $gdiff a a$ , that is, the identity patch for  $a$ , has cost 0 whereas  $Dμ-ins$  and  $Dμ-del$  will have strictly positive cost.

Since we mentioned the *identity patch* for an object, we might already introduce the two *special patches* that we need before talking about *cost*.

**The Identity Patch** Given the definition of  $gdiff$ , it is not hard to see that whenever  $x \equiv y$ , we produce a patch without any  $D-setl$ ,  $D-setr$ ,  $Dμ-ins$  or  $Dμ-del$ , let us call these the *change-introduction* constructors. Well, one can spare the comparisons made by  $gdiff$  and trivially define the identity patch for an object  $x$ ,  $gdiff-id x$ , by induction on  $x$ . A good sanity check, however, is to prove that this intuition is in fact correct:

```

gdiff-id-correct
: {n : ℕ} {t : Tel n} {ty : U n}
  → (a : EIU ty t) → gdiff-id a ≡ gdiff a a

```

**The Inverse Patch** If a patch  $gdiff x y$  is not the identity, then it has *change-introduction* constructors. If we swap every  $D-setl$  for  $D-setr$ , and  $Dμ-ins$  for  $Dμ-del$  and vice-versa, we get a patch that transforms  $y$  into  $x$ . We shall call this operation the inverse of a patch.

```

D-inv : {n : ℕ} {t : Tel n} {ty : U n}
  → Patch t ty → Patch t ty

```

As one would expect,  $gdiff y x$  or  $D-inv (gdiff x y)$  should be the same patch. Proving they are actually equal is very tricky and requires a lot of extra machinery regarding  $_⊔μ_$ , namely it needs to be associative and commutative. Moreover, inverses must distribute over it, that is  $D-inv (p_a ⊔μ p_b) = D-inv p_a ⊔μ D-inv p_b$ . Note that the problem only arises when we are evaluating  $p_a ⊔μ p_b$

for  $p_a$  and  $p_b$  with the same cost. We do prove, however, that  $gdiff y x$  and  $D-inv (gdiff x y)$  behave on the same way. This means that for all practical purposes, they are indistinguishable. We shall see what this statement means precisely in section 2.4.

### 2.3.1 The Cost Function

As we mentioned earlier, the cost function is one of the key pieces of the diff algorithm. It should assign a natural number to patches.

```

cost : {n : ℕ} {t : Tel n} {ty : U n} → Patch t ty → ℕ

```

The question is, how should we do this? The cost of transforming  $x$  and  $y$  intuitively leads one to think about *how far is x from y*. We see that the cost of a patch should not be too different from the notion of distance.

$$\text{dist } x y = \text{cost } (gdiff x y)$$

In order to achieve a meaningful definition, we will impose that our *cost* function must make the distance we defined above into a metric, that is:

$$\text{dist } x y = 0 \iff x = y \quad (1)$$

$$\text{dist } x y = \text{dist } y x \quad (2)$$

$$\text{dist } x y + \text{dist } y z \geq \text{dist } x z \quad (3)$$

Equation (1) tells that the cost of not changing anything must be 0, therefore the cost of every *non-change-introduction* constructor should be 0. The identity patch then has cost 0 by construction, as we seen it is exactly the patch with no *change-introduction* constructor.

Equation (2), on the other hand, tells that it should not matter whether we go from  $x$  to  $y$  or from  $y$  to  $x$ , the effort is the same. In patch-space, this means that the inverse of a patch should preserve its cost. Well, the inverse operation leaves everything unchanged but flips the *change-introduction* constructors to their dual counterpart. We will hence assign a cost  $c_{\oplus} = \text{cost } D-setl = \text{cost } D-setr$  and  $c_{\mu} = \text{cost } Dμ-ins = \text{cost } Dμ-del$  and guarantee this by construction already. Some care must be taken however, as if we define  $c_{\mu}$  and  $c_{\oplus}$  as constants we will say that inserting a tiny object has the same cost of inserting a gigantic object! that is not what we are looking for in a fine-tuned diff algorithm. Let us then define  $c_{\oplus} x y = \text{cost } (D-setr x y) = \text{cost } (D-setl x y)$  and  $c_{\mu} x = \text{cost } (Dμ-ins x) = \text{cost } (Dμ-del x)$  so we can take this fine-tuning into account.

Equation (3) is concerned with composition of patches. The aggregate cost of changing  $x$  to  $y$ , and then  $y$  to  $z$  should be greater than or equal to changing  $x$  directly to  $z$ . We do not have a composition operation over patches yet, but we can see that this is trivially satisfied. Let us denote the number of *change-introduction* constructors in a patch  $p$  by  $\#p$ . In the best case scenario,  $\#(gdiff x y) + \#(gdiff y z) = \#(gdiff x z)$ , this is the situation in which the changes of  $x$  to  $y$  and from  $y$  to  $z$  are non-overlapping. If they are overlapping, then some changes made from  $x$  to  $y$  must be changed again from  $y$  to  $z$ , yielding  $\#(gdiff x y) + \#(gdiff y z) > \#(gdiff x z)$ .

From equations (1) and (2) and from our definition of the equality patch and the inverse of a patch we already have that:

```

gdiff-id-cost : {n : ℕ} {t : Tel n} {ty : U n}
  → (a : EIU ty t) → cost (gdiff-id a) ≡ 0

```

```

D-inv-cost : {n : ℕ} {t : Tel n} {ty : U n}
  → (d : Patch t ty)
  → cost d ≡ cost (D-inv d)

```

In order to finalize our definition, we just need to find an actual value for  $c_{\oplus}$  and  $c_{\mu}$ . Let us then take a look at where the difference between these values comes into play. Assume we have stoped execution of `gdiffl` at the  $d_1 \sqcup_{\mu} d_2 \sqcup_{\mu} d_3$  expression. Here we have three patches to choose from:

$$\begin{aligned} d_1 &= \text{D}\mu\text{-ins } hdY :: \text{gdiffl } (x :: xs) (chY \# ys) \\ d_2 &= \text{D}\mu\text{-del } hdX :: \text{gdiffl } (chX \# xs) (y :: ys) \\ d_3 &= \text{D}\mu\text{-dwn } (\text{gdiffl } hdX hdY) \\ &:: \text{gdiffl } (chX \# xs) (chY \# ys) \end{aligned}$$

There is a catch here! Let us compare  $d_1$  and  $d_3$ . By choosing  $d_1$ , we would be opting to insert  $hdY$  instead of transforming  $hdX$  into  $hdY$ , this is indeed preferable if we dont have to delete  $hdX$  later on, in `gdiffl`  $(x :: xs) (chY \# ys)$ , as that would be a waste of information. This wasteful scenario happens when  $hdX \in chY \# ys$ . Now, assuming without loss of generality that  $hdX$  is the first element in  $chY \# ys$ , we have that:

$$d_1 = \text{D}\mu\text{-ins } hdY :: \text{D}\mu\text{-del } hdX :: \text{tail } d_3$$

Hence,  $\text{cost } d_1$  is  $c_{\mu} hdX + c_{\mu} hdY + w$ , for some  $w \in \mathbb{N}$ . Since we want to apply this to Haskell datatypes by the end of the day, it is acceptable to assume that  $hdX$  and  $hdY$  are values of the same finitary coproduct, representing the constructors of the fixed-point datatype. If  $hdX$  and  $hdY$  comes from different constructors of our fixed-point datatype in question, then  $hdX = i_j x'^3$  and  $hdY = i_k y'$  where  $j \neq k$ . The patch from  $hdX$  to  $hdY$  will therefore involve a `D-setl`  $x' y'$  or a `D-setr`  $y' x'$ , hence the cost of  $d_3$  becomes  $c_{\oplus} x' y' + w$ . Remember that in this situation it is wise to delete and insert instead of recursively changing. Since things are coming from a different constructors the structure of the outermost type is definitely changing, we want to reflect that! This means we need to select  $d_1$  instead of  $d_3$ , hence:

$$\Leftrightarrow \begin{aligned} c_{\mu} (i_j x') + c_{\mu} (i_k y') + w &< c_{\oplus} (i_j x') (i_k y') + w \\ c_{\mu} (i_j x') + c_{\mu} (i_k y') &< c_{\oplus} (i_j x') (i_k y') \end{aligned}$$

If  $hdX$  and  $hdY$  come from the same constructor, on the other hand, the story is slightly different. We now have  $hdX = i_j x'$  and  $hdY = i_j y'$ , the cost of  $d_1$  still is  $c_{\mu} (i_j x') + c_{\mu} (i_k y') + w$  but the cost of  $d_3$  is  $\text{dist } x' y' + w$ , since `gdiffl`  $hdX hdY$  will be `gdiffl`  $x' y'$  preceded by a sequence of `D-inr` and `D-inl` since  $hdX$  and  $hdY$  they come from the same coproduct injection, but these have cost 0. This is the situation that selecting  $d_3$  might be a wise choice, therefore we need:

$$\Leftrightarrow \begin{aligned} \text{dist } x' y' + w &< c_{\mu} (i_j x') + c_{\mu} (i_k y') + w \\ \text{dist } x' y' &< c_{\mu} (i_j x') + c_{\mu} (i_k y') \end{aligned}$$

In order to enforce this behaviour on our diffing algorithm, we need to assign values to  $c_{\mu}$  and  $c_{\oplus}$  that respects:

$$\text{dist } x' y' < c_{\mu} (i_j x') + c_{\mu} (i_k y') < c_{\oplus} (i_j x') (i_k y')$$

Now is the time for assigning values to  $c_{\mu}$  and  $c_{\oplus}$  and the diffing algorithm is completed. In fact, we define:

```
cost (D-A ()) = 0
cost (D-void) = 0
cost (D-inl d) = cost d
cost (D-inr d) = cost d
cost (D-setl xa xb) = 2 * (sizeE|U xa + sizeE|U xb)
cost (D-setr xa xb) = 2 * (sizeE|U xa + sizeE|U xb)
cost (D-pair da db) = cost da + cost db
cost (D-β d) = cost d
cost (D-top d) = cost d
cost (D-pop d) = cost d
cost (D-mu l) = costL l
```

```
costμ : {n : ℕ} {t : Tel n} {ty : U (suc n)}
  → Dμ ⊥p t ty → ℕ
costμ (Dμ-A ()) = 0
costμ (Dμ-ins x) = 1 + sizeE|U x
costμ (Dμ-del x) = 1 + sizeE|U x
costμ (Dμ-dwn x) = cost x
```

Where  $\text{costL} = \text{sum} \cdot \text{map } \text{cost}\mu$  and the size of an element is defined by:

```
sizeE|U : {n : ℕ} {t : Tel n} {u : U n} → E|U u t → ℕ
sizeE|U void = 1
sizeE|U (inl el) = 1 + sizeE|U el
sizeE|U (inr el) = 1 + sizeE|U el
sizeE|U (ela , elb) = sizeE|U ela + sizeE|U elb
sizeE|U (top el) = sizeE|U el
sizeE|U (pop el) = sizeE|U el
sizeE|U (mu el)
  = let (hdE , chE) = μ-open (mu el)
    in sizeE|U hdE + foldr _+_ 0 (map sizeE|U chE)
sizeE|U (red el) = sizeE|U el
```

Note that there are an infinite amount of definitions that would fit here. This is indeed a central topic of future work, as the `cost` function is what drives the diffing algorithm. This is deceptively complicated, though. Since  $\sqcup_{\mu}$  is not commutative nor associative, in general, we do not have much room to reason about the result of a `gdiffl`. A more careful definition of  $\sqcup_{\mu}$  that can provide more properties is paramount for a more careful study of the `cost` function. Defining  $\sqcup_{\mu}$  in such a way that it gives us a lattice over patches and still respects patch inverses is very tricky. One option would be to aim for a quasi-metric  $d$  instead of a metric (dropping equation (2)), this way inverses need not to distribute over  $\sqcup_{\mu}$  and we can still define a metric  $q$ , but now with codomain  $\mathbb{Q}$ , as  $q \ x \ y = \frac{1}{2} (d \ x \ y - d \ y \ x)$ .

## 2.4 Applying Patches

At this stage we are able to: work generically on a suitable universe; describe how elements of this universe can change and compute those changes. In order to make our framework usefull, though, we need to be able to apply the patches we compute. To our luck, the application of patches is easy, for we will only show the implementation for coproducts and fixedpoints here. The rest is very straight forward.

<sup>3</sup> We use  $i_j$  to denote the  $j$ -th injection into a finitary coproduct.



```

gapply : {n : ℕ}{t : Tel n}{ty : U n}
  → Patch t ty → EIU ty t → Maybe (EIU ty t)
gapply (D-inl diff) (inl el) = inl <M> gapply diff el
gapply (D-inr diff) (inr el) = inr <M> gapply diff el
gapply (D-setl x y) (inl el) with x  $\stackrel{?}{=}$  U el
... | yes _ = just (inr y)
... | no _ = nothing
gapply (D-setr y x) (inr el) with y  $\stackrel{?}{=}$  U el
... | yes _ = just (inl x)
... | no _ = nothing
gapply (D-setr _ _) (inl _) = nothing
gapply (D-setl _ _) (inr _) = nothing
gapply (D-inl diff) (inr el) = nothing
gapply (D-inr diff) (inl el) = nothing
gapply {ty =  $\mu$  ty} (D-mu d) el = gapplyL d (el :: [])  $\gg$  safeHead
:
gapplyL : {n : ℕ}{t : Tel n}{ty : U (suc n)}
  → Patch  $\mu$  t ty → List (EIU ( $\mu$  ty) t) → Maybe (List (EIU ( $\mu$  ty) t))
gapplyL [] [] = just []
gapplyL [] _ = nothing
gapplyL (D-mu A ()) :: []
gapplyL (D-mu-ins x :: d) l = gapplyL d l  $\gg$  glns x
gapplyL (D-mu-del x :: d) l = gDel x l  $\gg$  gapplyL d
gapplyL (D-mu-dwn dx :: d) [] = nothing
gapplyL (D-mu-dwn dx :: d) (y :: l) with  $\mu$ -open y
... | hdY , chY with gapply dx hdY
... | nothing = nothing
... | just y' = gapplyL d (chY ++ l)  $\gg$  glns y'

```

Where  $\langle M \rangle$  is the applicative-style application for the *Maybe* monad;  $\gg$  is the usual bind for the *Maybe* monad and *safeHead* is the partial head function with type  $[a] \rightarrow \text{Maybe } a$ . In *gapplyL*, we have a *glns* function, which will get a head and a list of children of a fixed point, will try to  $\mu$ -close it and add the result to the head of the remaining list. On the other hand, *gDel* will  $\mu$ -open the first element of the received list, compare it with the current head and return the tail of the input list appended to its children.

The important part of application is that it must produce the expected result. A correctness result guarantees that. Its proof is too big to be shown here, however, it has type:

```

correctness : {n : ℕ}{t : Tel n}{ty : U n}
  → (a b : EIU ty t)
  → gapply (gdiff a b) a  $\equiv$  just b

```

Also relevant is the fact that the inverse of a patch behaves as it should:

```

D-inv-sound
: {n : ℕ}{t : Tel n}{ty : U n}
  → (a b : EIU ty t)
  → gapply (D-inv (gdiff a b)) b  $\equiv$  just a


```

We have given algorithms for computing and applying differences over elements of a generic datatype. Moreover, we proved our algorithms are correct with respect to each other. This functionality is necessary for constructing a version control system, but it is by no means sufficient!

### 3. Patch Propagation

Let's say Bob and Alice perform edits in a given object, which are captured by patches  $p$  and  $q$ . In the version control setting, the natural question to ask is *how do we join these changes*.

There are two solutions that could possibly arise from this question. Either we group the changes made by  $p$  and by  $q$  (as long as they are compatible) and create a new patch to be applied on the source object, or, we calculate how to propagate the changes of  $p$  over  $q$  and vice-versa. Figure 4 illustrates these two options.



**Figure 4.** Residual Square on the left; three-way-merging on the right

The residual  $p/q$  of two patches  $p$  and  $q$  only makes sense if both  $p$  and  $q$  are aligned, that is, are defined for the same input. It captures the notion of incorporating the changes made by  $p$  in an object that has already been modified by  $q$ .

We chose to use the residual notion, as it seems to have more structure into it. Not to mention we could define  $p \cup q \equiv (q p) \cdot p \equiv (p/q) \cdot q$ . Unfortunately, however, there exists conflicts we need to take care of, which makes everything more complicated.

In an ideal world, we would expect the residual function to have type  $D a \rightarrow D a \rightarrow \text{Maybe } (D a)$ , where the partiality comes from receiving two non-aligned patches.

But what if Bob and Alice changes the same cell in their CSV file? Then it is obvious that someone (human) have to chose which value to use in the final, merged, version.

For this illustration, we will consider the conflicts that can arise from propagating the changes Alice made over the changes already made by Bob, that is,  $p_{\text{alice}}/p_{\text{bob}}$ .

- If Alice changes  $a_1$  to  $a_2$  and Bob changed  $a_1$  to  $a_3$ , with  $a_2 \neq a_3$ , we have an *update-update* conflict;
- If Alice adds information to a fixed-point, this is a *grow-left* conflict;
- When Bob added information to a fixed-point, which Alice didn't, a *grow-right* conflict arises;
- If both Alice and Bob add different information to a fixed-point, a *grow-left-right* conflict arises;
- If Alice deletes information that was changed by Bob we have an *delete-update* conflict;
- Last but not least, if Alice changes information that was deleted by Bob we have an *update-delete* conflict.

Above we see two distinct conflict types. An *update-update* conflict has to happen on a coproduct type, whereas the rest are restricted to fixed-point types. In Agda,

```

data C : {n : N} → Tel n → U n → Set where
  UpdUpd : {n : N}{t : Tel n}{a b : U n}
    → EU (a ⊕ b) t → EU (a ⊕ b) t → EU (a ⊕ b) t
    → C t (a ⊕ b)
  DelUpd : {n : N}{t : Tel n}{a : U (suc n)}
    → ValU a t → ValU a t → C t (μ a)
  UpdDel : {n : N}{t : Tel n}{a : U (suc n)}
    → ValU a t → ValU a t → C t (μ a)
  GrowL : {n : N}{t : Tel n}{a : U (suc n)}
    → ValU a t → C t (μ a)
  GrowLR : {n : N}{t : Tel n}{a : U (suc n)}
    → ValU a t → ValU a t → C t (μ a)
  GrowR : {n : N}{t : Tel n}{a : U (suc n)}
    → ValU a t → C t (μ a)

```

- *Pijul has this notion of handling a merge as a pushout, but it uses the free co-completion of a rather simple category. This doesn't give enough information for structured conflict solving.*
- **BACK THIS UP!**

### 3.1 Incorporating Conflicts

In order to track down these conflicts we need a more expressive patch data structure. We exploit  $D$ 's parameter for that matter. This approach has the advantage of separating conflicting from conflict-free patches on the type level, guaranteeing that we can only *apply* conflict-free patches.

The type of our residual<sup>4</sup> operation is:

```

_/_ : {n : N}{t : Tel n}{ty : U n}
  → Patch t ty → Patch t ty → Maybe (D C t ty)

```

We reiterate that the partiality comes from the fact the residual is not defined for non-aligned patches. We chose to make a partial function instead of receiving a proof of alignment purely for practical purposes. Defining alignment for our patches is very complicated.

The attentive reader might have noticed a symmetric structure on conflicts. This is not at all by chance. In fact, we can prove that the residual of  $p/q$  have the same (modulo symmetry) conflicts as  $q/p$ . This proof goes in two steps. Firstly, *residual-symmetry* proves that the symmetric of the conflicts of  $p/q$  appear in  $q/p$ , but this happens modulo a function. We then prove that this function does not introduce any new conflicts, it is purely structural.

```

residual-symmetry-thm
  : {n : N}{t : Tel n}{ty : U n}{k : D C t ty}
  → (d1 d2 : Patch t ty)
  → d1 / d2 ≡ just k
  → Σ (D C t ty → D C t ty)
  (λ op → d2 / d1 ≡ just (D-map C-sym (op k)))

residual-sym-stable : {n : N}{t : Tel n}{ty : U n}{k : D C t ty}
  → (d1 d2 : Patch t ty)
  → d1 / d2 ≡ just k
  → forget <M> (d2 / d1) ≡ just (map (↓-map-↓ C-sym) (forget k))

```

Here  $\downarrow\text{-map-}\downarrow$  takes care of hiding type indexes and *forget* is the canonical function with type  $D A t ty \rightarrow \text{List } (\downarrow A)$ ,  $\downarrow\text{-}$  encapsulates the type indexes of the different  $A$ 's we might come across.

<sup>4</sup>Our residual operation does not form a residual as in the Term Rewriting System sense[2]. It might, however, satisfy interesting properties. This is left as future work for now

Now, we can compute both  $p/q$  and  $q/p$  at the same time. It also backs up the intuition that using residuals or patch commutation (as in darcs) is not significantly different.

This means that  $p/q$  and  $q/p$ , although different, have the same conflicts (up to symmetry). Hence, we can (informally) define a *merge strategy* as a function  $m : \forall \{t ty\} \rightarrow C t ty \rightarrow D \perp_p t ty$ , which can now be mapped over  $D C t ty$  pointwise on its conflicts. We end up with an object of type  $D (D \perp_p) t ty$ . This is not a problem, however, since the free-monad structure on  $D$  provides us with a multiplication  $\mu_D : D (D A) t ty \rightarrow D A t ty$ . Therefore,  $\text{merge}_1 m : D C t ty \xrightarrow{\mu_D \cdot D m} \text{Patch } t ty$  is the *merge tool* for removing the conflicts of a single patch.

Even though the last paragraph was a simple informal sketch, as we did not have enough time to figure out precisely how to fit our framework in theory, we can see how solving conflicts is just a matter of walking over the patch structure and removing them, one by one. This is unsurprising, and is exactly what we currently do manually when merge conflicts arise on any mature version control system. The interesting bit, however, is that different ways of walking the patch might give us more or less information to handle a conflict. This opens up room for interesting automatic solvers, and besides not having a formal theory (yet!), they've shown to be very promising in practice as we will illustrate in section 5.

### 3.2 A remark on Type Safety

There is one major flaw in our approach so far. Our  $D\text{-}\mu$  constructor admits the creation of ill-formed patches. Consider the following simple example:

```

nat : U 0
nat = μ (u1 ⊕ v1)

ill-patch : Patch t nil nat
ill-patch = D-mu (Dμ-ins (inr (top void))) :: []

prf : ∀ el → gaply ill-patch el ≡ nothing
prf el = refl

```

On *ill-patch* we try to insert a *suc* constructor, which require one recursive argument, but then we simply end the patch. Agda realizes that this patch will lead nowhere in an instant.

We stress that on this model, we strive for type-safety on the API level, that is, if the user does not manually edit the patches, they will be type-safe. This is a problem that is not very hard to fix, however. We would just require two additional indexes on  $D\mu$ , expressing how many elements it expects, and how many it produces. Changing the whole library to handle these type-safe patches might require some coding effort, as many internal manipulations are gonna have to change. In particular, the way we handle conflicts is going to change significantly. The conflict type will have to receive these indexes too. We leave this as future work.

## 4. A Category of Patches

- *Having patch composition and inversion we can design a version control system for a single file in a categorical setting.*
  - *Label each composition chain as a branch, let residuals do the merging after conflict resolution.*

### To Research!

- *Define patch composition, prove it makes a category.*
- *But then... does it make sense to compute the composition of patches?*



- *In a vcs setting, we always have the intermediate files that originated the patches, meaning that composition can be defined semantically by:  $\text{apply}(p \cdot q) \equiv \text{apply}q \circ \text{apply}p$ , where  $\circ$  is the Kleisli composition of  $+1$ .*
- *This gives me an immediate category... how usefull is it?*

## 5. A Haskell Prototype

### 5.1 The Interface

- *present the `Diffable` typeclass.*

### 5.2 Sums-of-Products and Fixed points

- *present the `HasAlg` and `HasSOP` classes.*
- *mention the memoization table for diffing fixed points.*

### 5.3 A more involved example

- *Show how refactoring works*

## 6. Sketching a Control Version System

- Different views over the same datatype will give different diffs.
- `newtype` annotations can provide a great bunch of control over the algorithm.
- Directories are just rosetrees...

## 7. Related Work

### To Research!

- *Check out the antidiagonal with more attention: <http://blog.sigfpe.com/2007/09/type-of-distinct-pairs.html>*
  - *ANS: Diffing and Antidiagonals are fundamentally different. The antidiagonal for a type  $T$  is a type  $X$  such that there exists  $X \rightarrow T^2$ . That is,  $X$  produces two **distinct**  $T$ 's, whereas a `diff` produces a  $T$  given another  $T$ !*

## 8. Conclusion

- This is what we take out of it.

## References

- [1] T. Altenkirch, C. McBride, and P. Morris. Generic programming with dependent types. In *Spring School on Datatype Generic Programming*. Springer-Verlag, 2006.
- [2] M. Bezem, J. Klop, R. de Vrijer, and Terese. *Term Rewriting Systems*. Cambridge Tracts in Theoretical Computer Science. Cambridge University Press, 2003.
- [3] P. Bille. A survey on tree edit distance and related problems. *Theor. Comput. Sci.*, 337:217–239, 2005.
- [4] D. S. Hirschberg. Algorithms for the longest common subsequence problem. *J. ACM*, 24(4):664–675, Oct. 1977.
- [5] E. Lempitsink, S. Leather, and A. Löb. Type-safe diff for families of datatypes. In *Proceedings of the 2009 ACM SIGPLAN Workshop on Generic Programming, WGP '09*, pages 61–72, New York, NY, USA, 2009. ACM.
- [6] A. Mestanogullari, S. Hahn, J. K. Arni, and A. Löb. Type-level web apis with servant: An exercise in domain-specific generic programming. In *Proceedings of the 11th ACM SIGPLAN Workshop on Generic Programming, WGP 2015*, pages 1–12, New York, NY, USA, 2015. ACM.
- [7] T. Rendel and K. Ostermann. Invertible syntax descriptions: Unifying parsing and pretty printing. *SIGPLAN Not.*, 45(11):1–12, 2010.
- [8] J. Shaw. boomerang. <http://hackage.haskell.org/package/boomerang>, 2014. Accessed: November 2015.
- [9] S. Tieleman. Formalisation of version control with an emphasis on tree-structured data. Master's thesis, Universiteit Utrecht, Aug. 2006.
- [10] M. Vassena. Svc, a prototype of a structure-aware version control system. Master's thesis, Universiteit Utrecht, 2015.