

Best Title in the Universe

42

Victor Cacciari Miraldo Wouter Swierstra

University of Utrecht

{v.cacciarimiraldo,w.s.swierstra} at uu.nl

Abstract

stuff

Categories and Subject Descriptors D.1.1 [look]: for—this

General Terms Haskell

Keywords Haskell

1. Introduction

The majority of version control systems handle patches in a non-structured way. They see a file as a list of lines that can be inserted, deleted or modified, with no regard to the semantics of that specific file. The immediate consequence of such design decision is that we, humans, have to solve a large number of conflicts that arise from, in fact, non conflicting edits. Implementing a tool that knows the semantics of any file we happen to need, however, is no simple task, specially given the plethora of file formats we see nowadays.

This can be seen from a simple example. Lets imagine Alice and Bob are iterating over a cake’s recipe. They decide to use a version control system and an online repository to keep track of their modifications.




Figure 1. Sample CSV files

Lets say that both Bob and Alice are happy with their independent changes and want to make a final recipe. The standard way to track differences between files is the `diff3` [FIXBIB] unis tool. Running `diff3 Alice.csv 0.csv Bob.csv` would result in the output presented in figure 2. Every tag `====` marks a difference. Three

locations follows, formatted as `file:line type`. The change type can be a *Change*, *Append* or *Delete*. The first one, says that file 1 (Alice.csv) has a change in line 2 (1:2c) which is `flour, 2 , cp`; and files 2 and 3 have different changes in the same line. The tag `====3` indicates that there is a difference in file 3 only. Files 1 and 2 should append what changed in file 3 (line 4).

```
====
1:2c
    flour, 2 , cp
2:2c
    flour, 1 , cp
3:2c
    cakeflour, 1 , cp
====3
1:3a
2:3a
3:4c
    sugar, 1 , tsp
```

Figure 2. Output from `diff3`

If we try to merge the changes, `diff3` will flag a conflict and therefore require human interaction to solve it, as we can see by the presence of the `====` indicator in its output. However, Alice’s and Bob’s edits, in figure 1 do *not* conflict, if we take into account the semantics of CSV files. Although there is an overlapping edit at line 1, the fundamental editing unit is the cell, not the line.

We propose a structural diff that is not only generic but also able to track changes in a way that the user has the freedom to decide which is the fundamental editing unit. Our work was inspired by [?] and [?]. We did extensive changes in order to handle structural merging of patches. We also propose extensions to this algorithm capable of detecting purely structural operations such as refactorings and cloning.

The paper begins by exploring the problem, generically, in the Agda [FIXBIB] language. Once we have a provably correct algorithm, the details of a Haskell implementation of generic diff’ing are sketched. To open ground for future work, we present a few extensions to our initial algorithm that could be able to detect semantical operations such as *cloning* and *swapping*.

Contributions

- We provide a type-indexed definition of a generic diff, over the universe of context-free types.
- We provide the base, proven to be correct, algorithms for computing and applying a diff generically.

Permission to make digital or hard copies of part or all of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than ACM must be honored. Abstracting with credit is permitted. To copy otherwise, to republish, to post on servers, or to redistribute to lists, contact the Owner/Author. Request permissions from permissions@acm.org or Publications Dept., ACM, Inc., fax +1 (212) 869-0481. Copyright held by Owner/Author. Publication Rights Licensed to ACM.

- A notion of residual is used to propagate changes of different diffs, hence providing a bare mechanism for merging and conflict resolution.
- We provide a Haskell prototype with advanced, user defined, automatic conflict solving strategies.

Background

- *Should we have this section? It could be nice to at least mention the edit distance problem and that in the untyped scenario, the best running time is of $O(n^3)$. Types should allow us to bring this time lower.*

2. Structural Diffing

Alice and Bob were both editing a CSV file which represents data that is isomorphic to $[[Atom\ String]]$, where $Atom\ a$ is a simple tag that indicates that as should be treated abstractly, that is, either they are equal or different, we will not open these values to check for structural changes.

As we are tracking differences, there are a few operations that are inherent to our domain, such as: inserting; deleting; copying and updating. When we say *structural diffing*, however, we add another option to this list. Now we will also be able to go down the structure of some object and inspect its parts. To illustrate this, let us take Alice's change as in figure 1, her changes to the file could be described, structurally, as:

- I) Copy the first line;
- II) Enter the second line;
 - i) Copy the first field;
 - ii) Enter the second field;
 - Update atom "1" for atom "2".
 - iii) Copy the third field;
- III) Copy the third line.
- IV) Finish.

In figure 3 we show the patch that corresponds to that.




Figure 3. Alice's Patch

Consider now Bob's structural changes to the CSV file¹. If you overlap both, you should notice that there is *Upd* operation on top of another. This was in fact expected given that Alice and Bob performed changes in disjoint parts of the CSV file.

- *Diffing and tree-edit distance are very closely related problems.*
- *This should go on background, though.*

To Research!

¹Exercise to the reader! Clue: the last two operations are $Ins\ ["sugar", "1", "tsp"]\ End$

- *The LCS problem is closely related to diffing. We want to preserve the LCS of two structures! How does our diffing relate? Does this imply maximum sharing?*

- *ANS: No! We don't strive for maximum sharing. We strive for flexibility and customization. See refactoring*

- *connect this section and the next*

2.1 Context Free Datatypes

Although our running example, of CSV files, has type $[[Atom\ String]]$, lists of a themselves are in fact the least fixed point of the functor $X \mapsto 1 + a \times X$. Which is a *context-free type*, in the sense of [?]. For it is constructed following the grammar CF of context free types with a deBruijn representation for variables.

$$CF ::= 1 \mid 0 \mid CF \times CF \mid CF + CF \mid \mu\ CF \mid \mathbb{N}$$

In Agda, the CF universe is defined by:

```
data U : N → Set where
  u0 : {n : N} → U n
  u1 : {n : N} → U n
  _⊕_ : {n : N} → U n → U n → U n
  _⊗_ : {n : N} → U n → U n → U n
  β  : {n : N} → U (suc n) → U n → U n
  μ  : {n : N} → U (suc n) → U n
  vl : {n : N} → U (suc n)
  wk : {n : N} → U n → U (suc n)
```

Here, β stands for type application; vl is the topmost variable in scope and wk ignores the topmost variable in scope. We could have used a *Fin* to identify variables, and have one instead of two constructors for variables, but that would trigger more complicated definitions later on.

We stress that one of the main objectives of this project is to release a solid diffing and merging tool, that can provide formal guarantees, written in Haskell. The universe of user-defined Haskell types is smaller than context free types; in fact, we have fixed-points of sums-of-products. Therefore, we should be able to apply the knowledge acquired in Agda directly in Haskell. In fact, we did so! With a few adaptations here and there, to make the type-checker happy, the Haskell code is almost a direct translation, and will be discussed in section 5.

Stating the language of our types is not enough. We need to specify its elements too, after all, they are the domain which we seek to define our algorithms for! Defining elements of fixed-point types make things a bit more complicated, check [?] for a more in-depth explanation of these details. Long story short, we have to use a decreasing *Telescope* to satisfy the termination checker. In Agda, the elements of U are defined by:

```

data EIU : {n : ℕ} → U n → Tel n → Set where
  void : {n : ℕ} {t : Tel n}
    → EIU u1 t
  inl  : {n : ℕ} {t : Tel n} {a b : U n}
    (x : EIU a t) → EIU (a ⊕ b) t
  inr  : {n : ℕ} {t : Tel n} {a b : U n}
    (x : EIU b t) → EIU (a ⊕ b) t
  _,_  : {n : ℕ} {t : Tel n} {a b : U n}
    → EIU a t → EIU b t → EIU (a ⊗ b) t
  top  : {n : ℕ} {t : Tel n} {a : U n}
    → EIU a t → EIU v1 (tcons a t)
  pop  : {n : ℕ} {t : Tel n} {a b : U n}
    → EIU b t → EIU (wk b) (tcons a t)
  mu   : {n : ℕ} {t : Tel n} {a : U (suc n)}
    → EIU a (tcons (μ a) t) → EIU (μ a) t
  red  : {n : ℕ} {t : Tel n} {F : U (suc n)} {x : U n}
    → EIU F (tcons x t)
    → EIU (β F x) t

```

The `Tel` index is the telescope in which to look for the instantiation of type-variables. A value $(v : \text{EIU } \{n\} \text{ ty } t)$ reads roughly as: a value of type ty with n variables, applied to n types t with at most $n - 1$ variables. We need this decrease of type variables to convince the termination checker that our code is ok. It's Agda definition is:

```

data Tel : ℕ → Set where
  tnil : Tel 0
  tcons : {n : ℕ} → U n → Tel n → Tel (suc n)

```

Let us see a simple example of how types and elements are defined in this framework. Consider that we want to encode the list $(u : []) :: [U]$, for U being the unit type with the single constructor u . We start by defining the type of lists, this is an element of U (`suc n`), which later lets us define an element of that type.

```

list : {n : ℕ} → U (suc n)
list = μ (u1 ⊕ wk v1 ⊗ v1)

myList : {n : ℕ} {t : Tel n} → EIU list (tcons u1 t)
myList = mu (inr (pop (top void) , top (mu (inl void))))

```

So far so good. We seem to have the syntax figured out. But which operations can we perform to these elements? As we shall see, this choice of universe turns out to be very expressive, providing a plethora of interesting operations. The first very useful concept is the decidability of generic equality[?].

```

_≐_ : {n : ℕ} {t : Tel n} {u : U n} (x y : EIU u t) → Dec (x ≐ y)

```

But only comparing things will not get us very far. We need to be able to inspect our elements generically. Things like getting the list of immediate children, or computing their arity, that is, how many children do they have, are very useful.

```

children : {n : ℕ} {t : Tel n} {a : U (suc n)} {b : U n}
  → EIU a (tcons b t) → List (EIU b t)

arity : {n : ℕ} {t : Tel n} {a : U (suc n)} {b : U n}
  → EIU a (tcons b t) → ℕ

```

The advantage of doing so in Agda, is that we can prove that our definitions are correct.

```

children-arity-lemma
  : {n : ℕ} {t : Tel n} {a : U (suc n)} {b : U n}
  → (x : EIU a (tcons b t))
  → length (children x) ≐ arity x

```

We can even go a step further and say that every element is defined by a constructor and a vector of children, with the correct arity. This lets us treat generic elements as elements of a (typed) rose-tree, whenever this is convenient.

```

unplug : {n : ℕ} {t : Tel n} {a : U (suc n)} {b : U n}
  → (el : EIU a (tcons b t))
  → Σ (EIU a (tcons u1 t)) (λ x → Vec (EIU b t) (arity x))

plug : {n : ℕ} {t : Tel n} {a : U (suc n)} {b : U n}
  → (el : EIU a (tcons u1 t))
  → Vec (EIU b t) (arity el)
  → EIU a (tcons b t)

plug-correct : {n : ℕ} {t : Tel n} {a : U (suc n)} {b : U n}
  → (el : EIU a (tcons b t))
  → el ≐ plug (p1 (unplug el)) (p2 (unplug el))

```

• Vassena's and Loh's universe is the typed rose-tree! Correlate!!

This repertoire of operations, and the hability to inspect an element structurally, according to its type, gives us the toolset we need in order to start describing differences between elements. That is, we can now start discussing what does it mean to *diff* two elements or *patch* an element according to some description of changes.

2.2 Patches over a Context Free Type

A patch over T is an object that describe possible changes that can be made to objects of type T . The high-level idea is that diffing two objects $t_1, t_2 : T$ will produce a patch over T , whereas applying a patch over A to an object will produce a *Maybe T*. It is interesting to note that application can not be made total. Let's consider $T = X + Y$, and now consider a patch $(Left\ x) \xrightarrow{p} (Left\ x')$. What should be the result of applying p to a $(Right\ y)$? It is undefined!

The type of *diff*'s is defined by `D`. It is indexed by a type and a telescope, which is the same as saying that we only define *diff*'s for closed types². However, it also has a parameter A , this will be addressed later.

```

data D {a} (A : {n : ℕ} → Tel n → U n → Set a)
  : {n : ℕ} → Tel n → U n → Set a where

```

As we mentioned earlier, we are interested in analyzing the set of possible changes that can be made to objects of a type T . These changes depend on the structure of T , for the definition follows by induction on it.

For T being the Unit type, we can not modify it.

```

D-void : {n : ℕ} {t : Tel n} → D A t u1

```

For T being a product, we need to provide *diffs* for both its components.

```

D-pair : {n : ℕ} {t : Tel n} {a b : U n}
  → D A t a → D A t b → D A t (a ⊗ b)

```

For T being a coproduct, things become slightly more interesting. There are four possible ways of modifying a coproduct, which are defined by:

²Types that do not have any free type-variables

$\text{D-inl} : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{a b : \mathbb{U} n\}$
 $\rightarrow \text{D } A \text{ } t \text{ } a \rightarrow \text{D } A \text{ } t \text{ } (a \oplus b)$
 $\text{D-inr} : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{a b : \mathbb{U} n\}$
 $\rightarrow \text{D } A \text{ } t \text{ } b \rightarrow \text{D } A \text{ } t \text{ } (a \oplus b)$
 $\text{D-setl} : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{a b : \mathbb{U} n\}$
 $\rightarrow \text{E!U } a \text{ } t \rightarrow \text{E!U } b \text{ } t \rightarrow \text{D } A \text{ } t \text{ } (a \oplus b)$
 $\text{D-setr} : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{a b : \mathbb{U} n\}$
 $\rightarrow \text{E!U } b \text{ } t \rightarrow \text{E!U } a \text{ } t \rightarrow \text{D } A \text{ } t \text{ } (a \oplus b)$

We also need some housekeeping definitions to make sure we handle all types defined by \mathbb{U} .

$\text{D-}\beta : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{F : \mathbb{U} (\text{succ } n)\}\{x : \mathbb{U} n\}$
 $\rightarrow \text{D } A \text{ } (t \text{ cons } x \text{ } t) \text{ } F \rightarrow \text{D } A \text{ } t \text{ } (\beta F x)$
 $\text{D-top} : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{a : \mathbb{U} n\}$
 $\rightarrow \text{D } A \text{ } t \text{ } a \rightarrow \text{D } A \text{ } (t \text{ cons } a \text{ } t) \text{ } \text{v!}$
 $\text{D-pop} : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{a b : \mathbb{U} n\}$
 $\rightarrow \text{D } A \text{ } t \text{ } b \rightarrow \text{D } A \text{ } (t \text{ cons } a \text{ } t) \text{ } (\text{wk } b)$

Fixed points are handled by a list of *edit operations*. We will discuss them in detail later on.

$\text{D-mu} : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{a : \mathbb{U} (\text{succ } n)\}$
 $\rightarrow \text{List } (\text{D}\mu A \text{ } t \text{ } a) \rightarrow \text{D } A \text{ } t \text{ } (\mu a)$

The aforementioned parameter A goes is used in a single constructor, allowing us to have a free-monad structure over D 's. This shows to be very usefull for adding extra information, as we shall discuss, on section 3.1, for adding conflicts.

$\text{D-A} : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{ty : \mathbb{U} n\} \rightarrow A \text{ } t \text{ } ty \rightarrow \text{D } A \text{ } t \text{ } ty$

Finally, we define $\text{Patch } t \text{ } ty$ as $\text{D } (\lambda _ _ \rightarrow \perp) \text{ } t \text{ } ty$. Meaning that a Patch is a D with *no* extra information.

2.3 Producing Patches

Given a generic definition of possible changes, the primary goal is to produce an instance of this possible changes, for two specific elements of a type T . We shall call this process *diffing*. It is important to note that our gdiff function expects two elements of the same type! This contrasts with the work done by Vassena[?] and Lempink[?], where their diff takes objects of two different types.

For types which are not fixed points, the gdiff functions looks like:

$\text{gdiff} : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{ty : \mathbb{U} n\}$
 $\rightarrow \text{E!U } ty \text{ } t \rightarrow \text{E!U } ty \text{ } t \rightarrow \text{Patch } t \text{ } ty$
 $\text{gdiff} \{ty = \text{v!}\} (\text{top } a) (\text{top } b) = \text{D-top } (\text{gdiff } a \text{ } b)$
 $\text{gdiff} \{ty = \text{wk } u\} (\text{pop } a) (\text{pop } b) = \text{D-pop } (\text{gdiff } a \text{ } b)$
 $\text{gdiff} \{ty = \beta F x\} (\text{red } a) (\text{red } b) = \text{D-}\beta (\text{gdiff } a \text{ } b)$
 $\text{gdiff} \{ty = \text{u1}\} \text{void void} = \text{D-void}$
 $\text{gdiff} \{ty = ty \otimes tv\} (ay, av) (by, bv)$
 $= \text{D-pair } (\text{gdiff } ay \text{ } by) (\text{gdiff } av \text{ } bv)$
 $\text{gdiff} \{ty = ty \oplus tv\} (\text{inl } ay) (\text{inl } by) = \text{D-inl } (\text{gdiff } ay \text{ } by)$
 $\text{gdiff} \{ty = ty \oplus tv\} (\text{inr } av) (\text{inr } bv) = \text{D-inr } (\text{gdiff } av \text{ } bv)$
 $\text{gdiff} \{ty = ty \oplus tv\} (\text{inl } ay) (\text{inr } bv) = \text{D-setl } ay \text{ } bv$
 $\text{gdiff} \{ty = ty \oplus tv\} (\text{inr } av) (\text{inl } by) = \text{D-setr } av \text{ } by$
 $\text{gdiff} \{ty = \mu ty\} a \text{ } b = \text{D-mu } (\text{gdiffL } (a :: []) (b :: []))$

Where the gdiffL takes care of handling fixed point values. The important remark here is that it operates over lists of elements, instead of single elements. This is due to the fact that the children of a fixed point element is a (possibly empty) list of fixed point elements.

Fixed Points have a fundamental difference over regular algebraic datatypes. They can grow or shrink arbitralily. We have to account for that when tracking differences between their elements.

As we mentioned earlier, the diff of a fixed point is defined by a list of *edit operations*.

$\text{data D}\mu \{a\} (A : \{n : \mathbb{N}\} \rightarrow \text{Tel } n \rightarrow \mathbb{U} n \rightarrow \text{Set } a)$
 $: \{n : \mathbb{N}\} \rightarrow \text{Tel } n \rightarrow \mathbb{U} (\text{succ } n) \rightarrow \text{Set } a \text{ where}$

Again, we have a constructor for adding *extra* information, which is ignored in the case of Patches .

$\text{D}\mu\text{-A} : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{a : \mathbb{U} (\text{succ } n)\}$
 $\rightarrow A \text{ } t \text{ } (\mu a) \rightarrow \text{D}\mu A \text{ } t \text{ } a$

But the interesting bits are the *edit operations* we allow, where $\text{Val } a \text{ } t = \text{E!U } a \text{ } (t \text{ cons u1 } t)$:

$\text{D}\mu\text{-ins} : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{a : \mathbb{U} (\text{succ } n)\}$
 $\rightarrow \text{Val } a \text{ } t \rightarrow \text{D}\mu A \text{ } t \text{ } a$
 $\text{D}\mu\text{-del} : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{a : \mathbb{U} (\text{succ } n)\}$
 $\rightarrow \text{Val } a \text{ } t \rightarrow \text{D}\mu A \text{ } t \text{ } a$
 $\text{D}\mu\text{-dwn} : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{a : \mathbb{U} (\text{succ } n)\}$
 $\rightarrow \text{D } A \text{ } (t \text{ cons u1 } t) \text{ } a \rightarrow \text{D}\mu A \text{ } t \text{ } a$

The reader familiar with [?] will notice that they are almost the same (adapted to our choice of universe), with two differences: our diff type is *less type-safe*, which will be discussed in section 3.2; and instead of copying, we introduce a new constructor, $\text{D}\mu\text{-dwn}$, which is responsible for traversing down the type-structure. Copying is modelled by $\text{D}\mu\text{-dwn } (\text{gdiff-id } x)$. The intuition is that for every object x there is a diff that does not change x , we will look into this on section 2.3.

Before we delve into diffing fixed point values, we need some specialization of our generic operations for fixed points. Given that $\mu X.F \text{ } X \approx F \text{ } 1 \times [\mu X.F \text{ } X]$, that is, any inhabitant of a fixed-point type can be seen as a non-recursive head and a list of recursive children. We then make a specialized version of the plug and unplug functions, which lets us open and close fixed point values.

$\text{Open}\mu : \{n : \mathbb{N}\} \rightarrow \text{Tel } n \rightarrow \mathbb{U} (\text{succ } n) \rightarrow \text{Set}$
 $\text{Open}\mu \text{ } t \text{ } ty = \text{E!U } ty \text{ } (t \text{ cons u1 } t) \times \text{List } (\text{E!U } (\mu ty) \text{ } t)$

$\mu\text{-open} : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{ty : \mathbb{U} (\text{succ } n)\}$
 $\rightarrow \text{E!U } (\mu ty) \text{ } t \rightarrow \text{Open}\mu \text{ } t \text{ } ty$

$\mu\text{-close} : \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{ty : \mathbb{U} (\text{succ } n)\}$
 $\rightarrow \text{Open}\mu \text{ } t \text{ } ty \rightarrow \text{Maybe } (\text{E!U } (\mu ty) \text{ } t \times \text{List } (\text{E!U } (\mu ty) \text{ } t))$

Although the plug and unplug uses vectors, to remain total functions, we drop that restriction and switch to lists instead, this way we can easily construct a fixed-point with the beginning of the list of children, and return the unused children, this will be very conveient when defining how patches are applied. Nevertheless, a soundness lemma guarantees the correct behaviour.

$\mu\text{-close-resp-arity}$
 $: \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{ty : \mathbb{U} (\text{succ } n)\}\{a : \text{E!U } (\mu ty) \text{ } t\}$
 $\{hdA : \text{E!U } ty \text{ } (t \text{ cons u1 } t)\}\{chA : \text{List } (\text{E!U } (\mu ty) \text{ } t)\}$
 $\rightarrow \mu\text{-open } a \equiv (hdA, chA)$
 $\rightarrow \mu\text{-close } (hdA, chA ++ l) \equiv \text{just } (a, l)$

We denote the first component of an *opened* fixed point by its *value*, or *head*; whereas the second component by its children. The diffing of fixed points, which was heavily inspired by [?], is then defined by:

```

gdiffL : {n : ℕ} {t : Tel n} {ty : U (suc n)}
  → List (EIO (μ ty) t) → List (EIO (μ ty) t) → Patchμ t ty
gdiffL [] [] = []
gdiffL [] (y :: ys) with μ-open y
... | hdY, chY = Dμ-ins hdY :: (gdiffL [] (chY ++ ys))
gdiffL (x :: xs) [] with μ-open x
... | hdX, chX = Dμ-del hdX :: (gdiffL (chX ++ xs) [])
gdiffL (x :: xs) (y :: ys)
= let
  hdX, chX = μ-open x
  hdY, chY = μ-open y
  d1 = Dμ-ins hdY :: (gdiffL (x :: xs) (chY ++ ys))
  d2 = Dμ-del hdX :: (gdiffL (chX ++ xs) (y :: ys))
  d3 = Dμ-dwn (gdiff hdX hdY) :: (gdiffL (chX ++ xs) (chY ++ ys))
in d1 ⊔μ d2 ⊔μ d3

```

The first three branches are simple. To transform $[]$ into $[]$, we do not need to perform any action; to transform $[]$ into $y :: ys$, we need to insert the respective values; and to transform $x :: xs$ into $[]$ we need to delete the respective values. The interesting case happens when we want to transform $x :: xs$ into $y :: ys$. Here we have three possible diffs that perform the required transformation. We want to choose the diff with the least cost between the three of them. That is precisely what $_ \sqcup_\mu _$ does. As we shall see in section 2.3.1, this notion of cost is very delicate. The idea, however, is simple. If the heads are equal we have $d_3 = \text{D}\mu\text{-dwn } (\text{gdiff } hdX \text{ } hdY) \dots \approx \text{Copy } hdX$. Forcing to copy equal things will happen by assuring that for all a , $\text{gdiff } a \text{ } a$ has cost 0 whereas $\text{D}\mu\text{-ins}$ and $\text{D}\mu\text{-del}$ have strictly positive cost.

The Identity Patch Given the definition of $\text{gdiff } x \text{ } y$, it is not hard to see that: whenever $x \equiv y$, we produce a patch without any D-setl , D-setr , $\text{D}\mu\text{-ins}$ or $\text{D}\mu\text{-del}$. Which allows one to trivially define a the identity patch for an object x , $\text{gdiff-id } x$, by induction on x . A good sanity check, however, is to prove that this intuition is in fact correct:

```

gdiff-id-correct
: {n : ℕ} {t : Tel n} {ty : U n}
  → (a : EIO ty t) → gdiff-id a ≡ gdiff a a

```

In fact, the example provided in figure 3 is a diff produced by our algorithm, with the constructors simplified to improve readability.

2.3.1 The Cost Function

As we mentioned earlier, the cost function is one of the key pieces of the diff algorithm. In fact, the definition of the cost function should allow one to define a non-trivial measure over the set of all elements of a datatype. Henceforth, the identity patch should have cost 0. If the identity patch is one without D-setl , D-setr , $\text{D}\mu\text{-ins}$ or $\text{D}\mu\text{-del}$, this means every other constructor should have cost 0. Let us see now how do we attribute a cost to these constructors.

- *We might actually give some formal developments here...*
- **BEGIN IGNORE**

Unfortunately, however, formally studying the **cost** function turns out to be extremely complicated, as not only the generic nature of patches encompasses a plethora of cost behaviors, but the semantics of the domain one is applying the diff to might also require a slightly different definition.

This section does not show any formal development about the **cost** function, as we leave this as future work. Nonetheless, we explain the intuition behind our actual definition.

- **END IGNORE**

The **cost** of a **Patch** is used only in the **gdiffL** function, in order to choose which path to follow when the heads are not equal, therefore cannot be copied. Let us assume we have stopped execution at the $d_1 \sqcup_\mu d_2 \sqcup_\mu d_3$ expression, in **gdiffL**. Here we have:

```

d1 = Dμ-ins hdY :: gdiffL (x :: xs) (chY ++ ys)
d2 = Dμ-del hdX :: gdiffL (chX ++ xs) (y :: ys)
d3 = Dμ-dwn (gdiff hdX hdY)
    :: gdiffL (chX ++ xs) (chY ++ ys)

```

If $hdX \equiv hdY$, then d_3 will be selected as expected. Let us now assume that $hdX \neq hdY$. Two possibilities arise: either they can come from the same coproduct injection, or they don't. That is, imagine $hdX, hdY : \text{EIO } (a \oplus b \oplus c) (\text{tcons } u1 \text{ } t)$, for some types a, b, c and telescope t , assuming that there are no more coproducts inside a, b and c , unless wrapped by a μ^3 . Saying that hdX and hdY come from the same coproduct injection is saying that both hdX and hdY come from either a, b or c wrapped in their particular injections, or, in Haskell's term, they come from the same constructor.

If hdX and hdY do not come from the same coproduct injection, then d_3 should not be selected, as in fact it would be trying to change the outer constructor of a datatype instead of traversing inside of it and changing its non-recursive contents. That is to say, in this scenario, we want that $\text{cost } d_3 > \text{cost } d_i$, for $i \in \{1, 2\}$. Where $\text{cost } d_3 = \text{cost } (\text{D-set } hdX \text{ } hdY) + k$ for some $k \in \mathbb{N}$.

On the other hand, if hdX and hdY do come from the same coproduct injection, then we want to preserve this injection and track the recursive changes, for we then want $\text{cost } d_3 < \text{cost } d_i$, for $i \in \{1, 2\}$. Here, $\text{cost } d_3 = \text{cost } (\text{gdiff } (i_j \text{ } a') \text{ } (i_j \text{ } b')) + k$, for some $k \in \mathbb{N}$ and $hdX = i_j \text{ } a'$ and $hdY = i_j \text{ } b'$, where i_j is the injection to the j -th constructor. Diffing hdX and hdY will yield a sequence of D-inl and D-inr on top of $\text{gdiff } a' \text{ } b'$. But D-inl and D-inr have cost 0, hence $\text{cost } (\text{gdiff } (i_j \text{ } a') \text{ } (i_j \text{ } b')) = \text{cost } (\text{gdiff } a' \text{ } b')$.

Since cost for d_1 and d_2 involve inserting and deleting hdX and hdY . From the intuition gathered above, we can say that we want to have D-set as the most expensive operation, inserting and deleting should lie in the middle and just traversing the structure must be very cheap.

In short, we want to prevent deleting hdX and inserting hdY whenever there is information that could be preserved. With this intuition and the fact that the identity patch should have zero cost in mind, we then define the cost function as:

³In fact, this is how types are structured in Haskell, as *sums-of-products*, which is why we make this assumptions for the following reasoning.

```

cost : {n : ℕ} {t : Tel n} {ty : U n} → Patch t ty → ℕ
cost (D-A ()) = 0
cost (D-void) = 0
cost (D-inl d) = cost d
cost (D-inr d) = cost d
cost (D-setl xa xb) = 2 * (sizeE|U xa + sizeE|U xb)
cost (D-setr xa xb) = 2 * (sizeE|U xa + sizeE|U xb)
cost (D-pair da db) = cost da + cost db
cost (D-β d) = cost d
cost (D-top d) = cost d
cost (D-pop d) = cost d
cost (D-mu l) = sum cost μ l

cost μ : {n : ℕ} {t : Tel n} {ty : U (suc n)} → Dμ ⊥p t ty → ℕ
cost μ (Dμ-A ()) = 1 + sizeE|U x
cost μ (Dμ-ins x) = 1 + sizeE|U x
cost μ (Dμ-del x) = 1 + sizeE|U x
cost μ (Dμ-dwn x) = cost x

```

Where the size of an element is defined by:

```

sizeE|U : {n : ℕ} {t : Tel n} {u : U n} → E|U u t → ℕ
sizeE|U void = 1
sizeE|U (inl el) = 1 + sizeE|U el
sizeE|U (inr el) = 1 + sizeE|U el
sizeE|U (ela, elb) = sizeE|U ela + sizeE|U elb
sizeE|U (top el) = sizeE|U el
sizeE|U (pop el) = sizeE|U el
sizeE|U (mu el)
  = let (hdE, chE) = μ-open (mu el)
    in sizeE|U hdE + foldr _+_ 0 (map sizeE|U chE)
sizeE|U (red el) = sizeE|U el

```

We reiterate that this is an informal definition with nothing but intuition backing it up so far. This is, however, a central point of our current research. Our intention is to iterate over the design of this function until $dist = \text{curry } (cost \circ \text{uncurry } gdiff)$ becomes a metric over the set of elements of a given datatype, that is:

$$\begin{aligned}
dist\ x\ y = 0 & \Leftrightarrow x = y \\
dist\ x\ y & = dist\ y\ x \\
dist\ x\ y + dist\ y\ z & \leq dist\ x\ z \\
0 & \leq dist\ x\ y
\end{aligned}$$

Moreover, the cost of the identity patch is in fact 0.

```

gdiff-id-cost : {n : ℕ} {t : Tel n} {ty : U n}
→ (a : E|U ty t) → cost (gdiff-id a) ≡ 0

```

2.4 Applying Patches

At this stage we are able to: work generically on a suitable universe; describe how elements of this universe can change and compute those changes. In order to make our framework usefull, though, we need to be able to apply the patches we compute. To our luck, the application of patches is easy, for we will only show the implementation for coproducts and fixedpoints here. The rest is very straight forward.

```

gapply : {n : ℕ} {t : Tel n} {ty : U n}
→ Patch t ty → E|U ty t → Maybe (E|U ty t)
gapply (D-inl diff) (inl el) = inl <M> gapply diff el
gapply (D-inr diff) (inr el) = inr <M> gapply diff el
gapply (D-setl x y) (inl el) with x ≐=U el
...| yes _ = just (inr y)
...| no _ = nothing
gapply (D-setr y x) (inr el) with y ≐=U el
...| yes _ = just (inl x)
...| no _ = nothing
gapply (D-setr _ _) (inl _) = nothing
gapply (D-setl _ _) (inr _) = nothing
gapply (D-inl diff) (inr el) = nothing
gapply (D-inr diff) (inl el) = nothing
gapply {ty = μ ty} (D-mu d) el = gapplyL d (el :: []) >=> safeHead
:
gapplyL : {n : ℕ} {t : Tel n} {ty : U (suc n)}
→ Patch μ t ty → List (E|U (μ ty) t) → Maybe (List (E|U (μ ty) t))
gapplyL [] [] = just []
gapplyL [] _ = nothing
gapplyL (Dμ-A () :: _) =
gapplyL (Dμ-ins x :: d) l = gapplyL d l >=> g|ns x
gapplyL (Dμ-del x :: d) l = gDel x l >=> gapplyL d
gapplyL (Dμ-dwn dx :: d) [] = nothing
gapplyL (Dμ-dwn dx :: d) (y :: l) with μ-open y
...| hdY, chY with gapply dx hdY
...| nothing = nothing
...| just y' = gapplyL d (chY ++ l) >=> g|ns y'

```

Where $<M>$ is the applicative-style application for the *Maybe* monad; $>=>$ is the usual bind for the *Maybe* monad and *safeHead* is the partial head function with type $[a] \rightarrow \text{Maybe } a$. In *gapplyL*, we have a *g|ns* function, which will get a head and a list of children of a fixed point, will try to μ -close it and add the result to the head of the remaining list. On the other hand, *gDel* will μ -open the first element of the received list, compare it with the current head and return the tail of the input list appended to its children.

The important part of application is that it must produce the expected result. A correctness result guarantees that. Its proof is too big to be shown here, however, it has type:

```

correctness : {n : ℕ} {t : Tel n} {ty : U n}
→ (a b : E|U ty t)
→ gapply (gdiff a b) a ≡ just b

```

We have given algorithms for computing and applying differences over elements of a generic datatype. Moreover, we proved our algorithms are correct with respect to each other. This functionality is necessary for constructing a version control system, but it is by no means sufficient!

3. Patch Propagation

Let's say Bob and Alice perform edits in a given object, which are captured by patches p and q . In the version control setting, the natural question to ask is *how do we join these changes*.

There are two solutions that could possibly arise from this question. Either we group the changes made by p and by q (as long as they are compatible) and create a new patch to be applied on the source object, or, we calculate how to propagate the changes of p over q and vice-versa. Figure 4 illustrates these two options.

The residual p/q of two patches p and q only makes sense if both p and q are aligned, that is, are defined for the same input. It




Figure 4. Residual Square on the left; three-way-merging on the right

captures the notion of incorporating the changes made by p in an object that has already been modified by q .

We chose to use the residual notion, as it seems to have more structure into it. Not to mention we could define $p \cup q \equiv (q p) \cdot p \equiv (p/q) \cdot q$. Unfortunately, however, there exists conflicts we need to take care of, which makes everything more complicated.

In an ideal world, we would expect the residual function to have type $D\ a \rightarrow D\ a \rightarrow \text{Maybe}\ (D\ a)$, where the partiality comes from receiving two non-aligned patches.

But what if Bob and Alice changes the same cell in their CSV file? Then it is obvious that someone (human) have to chose which value to use in the final, merged, version.

For this illustration, we will consider the conflicts that can arise from propagating the changes Alice made over the changes already made by Bob, that is, $p_{\text{Alice}}/p_{\text{Bob}}$.

- If Alice changes a_1 to a_2 and Bob changed a_1 to a_3 , with $a_2 \neq a_3$, we have an *update-update* conflict;
- If Alice adds information to a fixed-point, this is a *grow-left* conflict;
- When Bob added information to a fixed-point, which Alice didn't, a *grow-right* conflict arises;
- If both Alice and Bob add different information to a fixed-point, a *grow-left-right* conflict arises;
- If Alice deletes information that was changed by Bob we have an *delete-update* conflict;
- Last but not least, if Alice changes information that was deleted by Bob we have an *update-delete* conflict.

Above we see two distinct conflict types. An *update-update* conflict has to happen on a coproduct type, whereas the rest are restricted to fixed-point types. In Agda,

```
data C : {n : ℕ} → Tel n → U n → Set where
  UpdUpd : {n : ℕ}{t : Tel n}{a b : U n}
    → EU (a ⊕ b) t → EU (a ⊕ b) t → EU (a ⊕ b) t
    → C t (a ⊕ b)
  DelUpd : {n : ℕ}{t : Tel n}{a : U (suc n)}
    → ValU a t → ValU a t → C t (μ a)
  UpdDel : {n : ℕ}{t : Tel n}{a : U (suc n)}
    → ValU a t → ValU a t → C t (μ a)
  GrowL : {n : ℕ}{t : Tel n}{a : U (suc n)}
    → ValU a t → C t (μ a)
  GrowLR : {n : ℕ}{t : Tel n}{a : U (suc n)}
    → ValU a t → ValU a t → C t (μ a)
  GrowR : {n : ℕ}{t : Tel n}{a : U (suc n)}
    → ValU a t → C t (μ a)
```

- Pijul has this notion of handling a merge as a pushout, but it uses the free co-completion of a rather simple category. This doesn't give enough information for structured conflict solving.
- **BACK THIS UP!**

3.1 Incorporating Conflicts

In order to track down these conflicts we need a more expressive patch data structure. We exploit D 's parameter for that matter. This approach has the advantage of separating conflicting from conflict-free patches on the type level, guaranteeing that we can only *apply* conflict-free patches.

The type of our residual⁴ operation is:

$$\begin{aligned} _/_ &: \{n : \mathbb{N}\}\{t : \text{Tel } n\}\{ty : U\ n\} \\ &\rightarrow \text{Patch } t\ ty \rightarrow \text{Patch } t\ ty \rightarrow \text{Maybe } (D\ C\ t\ ty) \end{aligned}$$

We reiterate that the partiality comes from the fact the residual is not defined for non-aligned patches. We chose to make a partial function instead of receiving a proof of alignment purely for practical purposes. Defining alignment for our patches is very complicated.

The attentive reader might have noticed a symmetric structure on conflicts. This is not at all by chance. In fact, we can prove that the residual of p/q have the same (modulo symmetry) conflicts as q/p . This proof goes in two steps. Firstly, *residual-symmetry* proves that the symmetric of the conflicts of p/q appear in q/p , but this happens modulo a function. We then prove that this function does not introduce any new conflicts, it is purely structural.

```
residual-symmetry-thm
: {n : ℕ}{t : Tel n}{ty : U n}{k : D C t ty}
→ (d1 d2 : Patch t ty)
→ d1 / d2 ≡ just k
→ Σ (D C t ty → D C t ty)
  (λ op → d2 / d1 ≡ just (D-map C-sym (op k)))

residual-sym-stable : {n : ℕ}{t : Tel n}{ty : U n}{k : D C t ty}
→ (d1 d2 : Patch t ty)
→ d1 / d2 ≡ just k
→ forget <M> (d2 / d1) ≡ just (map (↓-map-↓ C-sym) (forget k))
```

Here $\downarrow\text{-map-}\downarrow$ takes care of hiding type indexes and *forget* is the canonical function with type $D\ A\ t\ ty \rightarrow \text{List } (\downarrow A)$, $\downarrow__$ encapsulates the type indexes of the different A 's we might come across.

Now, we can compute both p/q and q/p at the same time. It also backs up the intuition that using residuals or patch commutation (as in darcs) is not significantly different.

This means that p/q and q/p , although different, have the same conflicts (up to symmetry). Hence, we can (informally) define a *merge strategy* as a function $m : \forall \{t\ ty\} \rightarrow C\ t\ ty \rightarrow D\ \perp_p\ t\ ty$, which can now be mapped over $D\ C\ t\ ty$ pointwise on its conflicts. We end up with an object of type $D\ (D\ \perp_p)\ t\ ty$. This is not a problem, however, since the free-monad structure on D provides us with a multiplication $\mu_D : D\ (D\ A)\ t\ ty \rightarrow D\ A\ t\ ty$. Therefore, $\text{merge}_1\ m : D\ C\ t\ ty \xrightarrow{\mu_D \cdot D\ m} \text{Patch } t\ ty$ is the *merge tool* for removing the conflicts of a single patch.

Even though the last paragraph was a simple informal sketch, as we did not have enough time to figure out precisely how to fit our framework in theory, we can see how solving conflicts is just a matter of walking over the patch structure and removing them,

⁴Our residual operation does not form a residual as in the Term Rewriting System sense[?]. It might, however, satisfy interesting properties. This is left as future work for now

one by one. This is unsurprising, and is exactly what we currently do manually when merge conflicts arise on any mature CVS. The interesting bit, however, is that different ways of walking the patch might give us more or less information to handle a conflict. This opens up room for interesting automatic solvers, and besides not having a formal theory (yet!), they've shown to be very promising in practice as we will illustrate in section 5.

3.2 A remark on Type Safety

There is one major flaw in our approach so far. Our `D-mu` constructor admits the creation of ill-formed patches. Consider the following simple example:

```

nat : U 0
nat =  $\mu$  (u1  $\oplus$  v1)

ill-patch : Patch tnil nat
ill-patch = D-mu (D $\mu$ -ins (inr (top void))) :: []

prf :  $\forall$  el  $\rightarrow$  gapply ill-patch el  $\equiv$  nothing
prf el = refl

```

On `ill-patch` we try to insert a `suc` constructor, which require one recursive argument, but then we simply end the patch. Agda realizes that this patch will lead nowhere in an instant.

We stress that on this model, we strive for type-safety on the API level, that is, if the user does not manually edit the patches, they will be type-safe. This is a problem that is not very hard to fix, however. We would just require two additional indexes on `D μ` , expressing how many elements it expects, and how many it produces. Changing the whole library to handle these type-safe patches might require some coding effort, as many internal manipulations are gonna have to change. In particular, the way we handle conflicts is going to change significantly. The conflict type will have to receive these indexes too. We leave this as future work.

4. A Category of Patches

- *Having patch composition and inversion we can design a version control system for a single file in a categorical setting.*
 - *Label each composition chain as a branch, let residuals do the merging after conflict resolution.*

To Research!

- *Define patch composition, prove it makes a category.*
- *But then... does it make sense to compute the composition of patches?*
- *In a vcs setting, we always have the intermediate files that originated the patches, meaning that composition can be defined semantically by: $\text{apply}(p \cdot q) \equiv \text{apply}q \circ \text{apply}p$, where \circ is the Kleisli composition of $+1$.*
- *This gives me an immediate category... how usefull is it?*

5. A Haskell Prototype

5.1 The Interface

- *present the `Diffable` typeclass.*

5.2 Sums-of-Products and Fixed points

- *present the `HasAlg` and `HasSOP` classes.*
- *mention the memoization table for diffing fixed points.*

5.3 A more involved example

- *Show how refactoring works*

6. Sketching a Control Version System

- Different views over the same datatype will give different diffs.
- `newtype` annotations can provide a great bunch of control over the algorithm.
- Directories are just rosetrees...

7. Related Work

To Research!

- *Check out the antidiagonal with more attention: <http://blog.sigfpe.com/2007/09/type-of-distinct-pairs.html>*
 - *ANS: Diffing and Antidiagonals are fundamentally different. The antidiagonal for a type T is a type X such that there exists $X \rightarrow T^2$. That is, X produces two **distinct** T 's, whereas a `diff` produces a T given another T !*

8. Conclusion

- This is what we take out of it.